Analytic Properties of 2PI Approximation Schemes

Author: Michael Brown

Supplement to thesis Chapter 2

Mathematica notebook to compute solution of the zero dimensional toy model and compare perturbation theory, Pade, Borel-Pade, 2PI and hybrid 2PI-Pade approximations.

Classical action and solutions

```
Scl = 1/2 m^2 q^2 + 1/4! λ q^4;
```

The classical solutions extremize the action:

```
clsoln = Solve[D[Scl, q] == 0, q]
```

$$\left\{\{q \to 0\}, \left\{q \to -\frac{i\sqrt{6}\, m}{\sqrt{\lambda}}\right\}, \left\{q \to \frac{i\sqrt{6}\, m}{\sqrt{\lambda}}\right\}\right\}$$

The real solution is a minimum and the imaginary solutions are saddle points:

```
D[Scl, q, q] /. clsoln
```

$$\left\{m^2, -2\, m^2, -2\, m^2\right\}$$

with action:

```
Scl /. clsoln
```

$$\left\{0, -\frac{3\, m^4}{2\,\lambda}, -\frac{3\, m^4}{2\,\lambda}\right\}$$

```
ContourPlot[Abs[Exp[-Scl]] /. {m → 1, λ → 1, q → x + ⅈ y}, {x, -3, 3}, {y, -4, 4},
 PlotPoints → 30, Contours → Table[0.5 i, {i, 20}], PlotRange → {0, 10},
 FrameLabel → {"Re[q]", "Im[q]"}, PlotTheme → "Scientific", PlotLegends → Automatic,
 ContourShading → Automatic, ContourStyle → {Black, Dashed}, LabelStyle → Medium]
ContourPlot[Arg[Exp[-Scl]] /. {m → 1, λ → 1, q → x + ⅈ y}, {x, -3, 3},
 {y, -4, 4}, PlotPoints → 30, FrameLabel → {"Re[q]", "Im[q]"},
 Contours → 5, PlotTheme → "Scientific", PlotLegends → Automatic,
 ContourShading → Automatic, ContourStyle → {Black, Dashed}, LabelStyle → Medium]
```






```
(*Export["z-integrand-abs.pdf",%93]
 Export["z-integrand-arg.pdf",%94]*)
```

Partition function and connected generating function

```
ClearAll[normalization, z, w]
```

```
normalization = Integrate[Exp[-Scl], {q, -∞, ∞}, Assumptions → {m > 0, λ > 0}]
```

$$\frac{\sqrt{3}\; m\, e^{\frac{3 m^4}{4 \lambda}}\, K_{\frac{1}{4}}\!\left(\frac{3 m^4}{4 \lambda}\right)}{\sqrt{\lambda}}$$

$$\texttt{z[k\_, l\_] := z[k, l] = } \frac{1}{\texttt{normalization}}$$

$$\texttt{Integrate}\left[\texttt{Exp}\left[-\texttt{Scl} - \frac{1}{2}\,\texttt{k}\,\texttt{q}^2 - \frac{1}{4\,!}\,\texttt{l}\,\texttt{q}^4\right], \{\texttt{q}, -\infty, \infty\}, \texttt{Assumptions} \to \left\{\lambda + \texttt{l} > 0,\ \texttt{m}^2 + \texttt{k} > 0\right\}\right]$$

```
z[k, 0]
```

$$\frac{e^{\frac{3 (k+m^2)^2}{4 \lambda} - \frac{3 m^4}{4 \lambda}} \sqrt{\lambda \left(k + m^2\right)}\; K_{\frac{1}{4}}\!\left(\frac{3 \left(m^2+k\right)^2}{4 \lambda}\right)}{\sqrt{\lambda}\; m\, K_{\frac{1}{4}}\!\left(\frac{3 m^4}{4 \lambda}\right)}$$

```
w[k_, l_] := w[k, l] = -Log[z[k, l]]
```

$$\texttt{w[k, l] // FullSimplify}\left[\#, \texttt{Assumptions} \to \left\{\lambda + \texttt{l} > 0,\ \texttt{m}^2 + \texttt{k} > 0\right\}\right]\ \&$$

$$-\log\left(\frac{e^{\frac{3}{4}\left(\frac{(k+m^2)^2}{\lambda+l} - \frac{m^4}{\lambda}\right)} \sqrt{\frac{\lambda \left(k+m^2\right)}{\lambda+l}}\; K_{\frac{1}{4}}\!\left(\frac{3 \left(m^2+k\right)^2}{4 \left(l+\lambda\right)}\right)}{m\, K_{\frac{1}{4}}\!\left(\frac{3 m^4}{4 \lambda}\right)}\right)$$

Exact Propagator

```
Gexact = FullSimplify[2 D[w[k, 0], k] /. k → 0]
```

$$\frac{3 m^2 \left(\frac{K_{\frac{3}{4}}\left(\frac{3 m^4}{4 \lambda}\right)}{K_{\frac{1}{4}}\left(\frac{3 m^4}{4 \lambda}\right)} - 1\right)}{\lambda}$$

```
Plot[
 {Re[Gexact], Im[Gexact] /. λ → λ - 0.00001 ⅈ, Im[Gexact] /. λ → λ + 0.00001 ⅈ} /. m → 1 //
  Evaluate, {λ, -5, 5}, PlotLegends → {"Re[G̅]", "Im[G̅[λ-ⅈϵ]]", "Im[G̅[λ+ⅈϵ]]"},
 FrameLabel → {λ, G̅}, PlotTheme → "Scientific",
 PlotStyle → {Dashing[{}], Dashed, DotDashed}, LabelStyle → Medium]
```




```
(*Export["Gexact.pdf",%97]*)
```

```
ContourPlot[Abs[Gexact] /. {m → 1, λ → x + ⅈ y}, {x, -3, 3}, {y, -4, 4}, PlotPoints → 50,
 PlotRange → {0, 10}, FrameLabel → {"Re[λ]", "Im[λ]"}, PlotTheme → "Scientific",
 PlotLegends → Automatic, Contours → Table[i, {i, 0, 1.3, 1/10}], LabelStyle → Medium]
ContourPlot[Arg[Gexact] /. {m → 1, λ → x + ⅈ y}, {x, -3, 3},
 {y, -4, 4}, PlotPoints → 40, FrameLabel → {"Re[λ]", "Im[λ]"},
 PlotTheme → "Scientific", PlotLegends → Automatic, LabelStyle → Medium]
```






```
(*Export["Gexact-abs.pdf",%99]
 Export["Gexact-arg.pdf",%100]*)
```

Discontinuity of the cut in λ:

```
disc = Series[(Gexact /. λ → λ + ⅈ ϵ) - (Gexact /. λ → λ - ⅈ ϵ),
    {ϵ, 0, 0}, Assumptions → {m > 0, λ < 0, ϵ > 0}] // Normal //
  FullSimplify[#, Assumptions → {m > 0, λ < 0, ϵ > 0}] &
```

$$-\frac{8\sqrt{2}}{\pi m^2\left(I_{-\frac{1}{4}}\left(\frac{3 m^4}{4\lambda}\right)^2 - I_{\frac{1}{4}}\left(\frac{3 m^4}{4\lambda}\right)^2\right)}$$

Spectral function:

```
σexact = (Im[disc] /. λ → -λ)/(-2 π) HeavisideTheta[λ]
```

$$\frac{4\sqrt{2}\,\theta(\lambda)\,\mathrm{Im}\left(\frac{1}{m^2\left(I_{-\frac{1}{4}}\left(-\frac{3 m^4}{4\lambda}\right)^2 - I_{\frac{1}{4}}\left(-\frac{3 m^4}{4\lambda}\right)^2\right)}\right)}{\pi^2}$$

```
Plot[σexact /. m → 1, {λ, -1, 10}, FrameLabel → {λ, σ},
 PlotTheme → "Scientific", LabelStyle → Medium]
```




```
(*Export["sigma-exact.pdf",%103]*)
```

Exact Vertex Function

```
V4exact =
  Simplify[-(-4 D[w[k, 0], k, k] - 2 Gexact^2)/(Gexact^4) /. k → 0, Assumptions → {m > 0, λ > 0}];
```

```
Series[V4exact, {λ, 0, 2}]
```

$$\lambda - \frac{3\lambda^2}{2\,m^4} + O[\lambda]^{5/2}$$

Perturbation series

```
ClearAll[Gpert]
```

```
Gpert[n_] := Gpert[n] = Series[Gexact, {λ, 0, n}, Assumptions → {m > 0, λ > 0}] // Normal
```

```
Series[Gexact, {λ, 0, 1}, Assumptions → {m > 0, λ > 0}]
```

$$\frac{1}{m^2} - \frac{\lambda}{2\,m^6} + O\left(\lambda^{3/2}\right)$$

```
Gpert[2]
```

$$\frac{2\,\lambda^2}{3\,m^{10}} - \frac{\lambda}{2\,m^6} + \frac{1}{m^2}$$

```
Table[{n, Gpert[n]}, {n, 20}]; Gpert[20]
```

$$\frac{38\,131\,059\,969\,414\,309\,154\,\lambda^{20}}{59\,049\,m^{82}} - \frac{249\,615\,310\,568\,664\,912\,892\,639\,\lambda^{19}}{5\,159\,780\,352\,m^{78}} + \frac{75\,079\,607\,548\,893\,602\,\lambda^{18}}{19\,683\,m^{74}} - \frac{45\,494\,616\,421\,387\,671\,961\,\lambda^{17}}{143\,327\,232\,m^{70}} + \frac{20\,384\,334\,558\,514\,\lambda^{16}}{729\,m^{66}} - \frac{6\,944\,870\,083\,473\,751\,\lambda^{15}}{2\,654\,208\,m^{62}} + \frac{63\,444\,074\,282\,\lambda^{14}}{243\,m^{58}} - \frac{41\,657\,327\,595\,361\,\lambda^{13}}{1\,492\,992\,m^{54}} + \frac{2\,339\,780\,194\,\lambda^{12}}{729\,m^{50}} - \frac{33\,152\,545\,703\,\lambda^{11}}{82\,944\,m^{46}} + \frac{4\,394\,374\,\lambda^{10}}{81\,m^{42}} - \frac{18\,639\,247\,\lambda^{9}}{2304\,m^{38}} + \frac{108\,386\,\lambda^{8}}{81\,m^{34}} - \frac{858\,437\,\lambda^{7}}{3456\,m^{30}} + \frac{1418\,\lambda^{6}}{27\,m^{26}} - \frac{619\,\lambda^{5}}{48\,m^{22}} + \frac{34\,\lambda^{4}}{9\,m^{18}} - \frac{11\,\lambda^{3}}{8\,m^{14}} + \frac{2\,\lambda^{2}}{3\,m^{10}} - \frac{\lambda}{2\,m^{6}} + \frac{1}{m^{2}}$$

```
Plot[{Re[Gexact], Table[Gpert[n], {n, 0, 5}]} /. m → 1 // Evaluate,
 {λ, -1, 1}, PlotLegends → ({"Re[G̅]", Table[Gₙ, {n, 0, 5}]} // Flatten),
 FrameLabel → {λ, G}, PlotTheme → "Scientific", PlotRange → {1/2, 3/2},
 PlotStyle → ({Automatic, Dotted, Dashing[{Small, Medium}],
    Dashing[{0, Small, Tiny}], Dashing[Large], DotDashed}), LabelStyle → Medium]
```




```
(*Export["G-pert-series.pdf",%105]*)
```

Relative error shows that the series is asymptotic, not convergent:

```
ListLogPlot[Table[{n, Abs[Gpert[n]/Gexact - 1 /. {m → 1, λ → 1/4}]}, {n, 0, 20}] // N,
 Filling → Bottom, PlotRange → All,
 AxesLabel → {"n", "|(Gn-G̅)/G̅|"}, LabelStyle → Medium]
```




```
(*Export["G-pert-asympt-series.pdf",%107]*)
```

Pade Approximants

```
Pade[n_] := Pade[n] = PadeApproximant[Gexact, {λ, 0, {n, n + 1}}]
```

```
Table[{n, Pade[n]}, {n, 0, 4}] // TableForm
```

0	$\frac{1}{m^2\left(\frac{\lambda}{2 m^4}+1\right)}$
1	$\frac{\frac{2\lambda}{m^6}+\frac{1}{m^2}}{\frac{7\lambda^2}{12 m^8}+\frac{5\lambda}{2 m^4}+1}$
2	$\frac{\frac{59\lambda^2}{12 m^{10}}+\frac{16\lambda}{3 m^6}+\frac{1}{m^2}}{\frac{77\lambda^3}{72 m^{12}}+\frac{43\lambda^2}{6 m^8}+\frac{35\lambda}{6 m^4}+1}$
3	$\frac{\frac{15\lambda^3}{m^{14}}+\frac{155\lambda^2}{6 m^{10}}+\frac{10\lambda}{m^6}+\frac{1}{m^2}}{\frac{385\lambda^4}{144 m^{16}}+\frac{295\lambda^3}{12 m^{12}}+\frac{365\lambda^2}{12 m^8}+\frac{21\lambda}{2 m^4}+1}$
4	$\frac{\frac{7945\lambda^4}{144 m^{18}}+\frac{1190\lambda^3}{9 m^{14}}+\frac{315\lambda^2}{4 m^{10}}+\frac{16\lambda}{m^6}+\frac{1}{m^2}}{\frac{7315\lambda^5}{864 m^{20}}+\frac{14315\lambda^4}{144 m^{16}}+\frac{11935\lambda^3}{72 m^{12}}+\frac{259\lambda^2}{3 m^8}+\frac{33\lambda}{2 m^4}+1}$

```
Plot[{Re[Gexact], Table[Pade[n], {n, 0, 2}]} /. m → 1 // Evaluate, {λ, -3, 3},
 PlotLegends → ({"Re[G̅]", Table[Subsuperscript[Pade, n + 1, n], {n, 0, 2}]} // Flatten),
 FrameLabel → {λ, G}, PlotTheme → "Scientific", PlotRange → {0, 2},
 PlotStyle → {Automatic, Dashed, Dotted, DotDashed}, LabelStyle → Medium]
```




```
(*Export["Pade-approx.pdf",%109]*)
```

Pade–approx.pdf

```
discPade[n_, ϵ_] := (Pade[n] /. λ → λ + ⅈ ϵ) - (Pade[n] /. λ → λ - ⅈ ϵ) //
  FullSimplify[#, Assumptions → {m > 0, λ < 0, ϵ > 0}] &
```

```
discPade[0, ϵ]
```

$$-\frac{4\,i\,m^2\,\epsilon}{\left(\lambda + 2\,m^4\right)^2 + \epsilon^2}$$

Spectral function:

```
σPade[n_, ϵ_] := σPade[n, ϵ] = (Im[discPade[n, ϵ]] /. λ → -λ) / (-2 π) HeavisideTheta[λ]
```

```
σPade[0, 0.0001] /. m → 1
```

$$-\frac{\theta(\lambda)\,\mathrm{Im}\left(-\frac{0.+0.0001\,i}{0.25\,\lambda^2-1.\,\lambda+1.}\right)}{2\,\pi}$$

```
Plot[{σexact, Table[σPade[n, 0.00001], {n, 0, 3}]} /. m → 1 // Evaluate,
 {λ, 0, 10}, FrameLabel → {λ, σ}, PlotTheme → "Scientific",
 PlotLegends → ({"σ", Table[σ[Subsuperscript[Pade, n + 1, n]], {n, 0, 3}]} // Flatten),
 PlotRange → {0, 1/3}, PlotStyle → {Automatic, Dashed, Dotted, DotDashed},
 PlotPoints → 100, LabelStyle → Medium]
```




```
(*Export["sigma-Pade.pdf",%111]*)
```

sigma–Pade.pdf

Borel Pade

```
ClearAll[BorelSeriesCoeff, BorelPade, BorelTransf]

BorelSeriesCoeff[n_] := BorelSeriesCoeff[n] =
  (SeriesCoefficient[Gexact, {λ, 0, n}, Assumptions → {m > 0, λ > 0}] 1/n!)

BorelTransf[n_] := BorelTransf[n] = Sum[BorelSeriesCoeff[k] λ^k x^k, {k, 0, n}]

BorelPade[a_, b_] :=
 BorelPade[a, b] = PadeApproximant[BorelTransf[a + b], {x, 0, {a, b}}]

Table[BorelPade[a, b], {a, 3}, {b, 3}] // Simplify
```

$$\begin{pmatrix} \frac{6 m^4+x \lambda}{6 m^6+4 x \lambda m^2} & \frac{6 m^2 \left(4 m^4+x \lambda\right)}{24 m^8+18 x \lambda m^4+x^2 \lambda^2} & \frac{48 m^6 \left(9 m^4+2 x \lambda\right)}{432 m^{12}+312 x \lambda m^8+12 x^2 \lambda^2 m^4+x^3 \lambda^3} \\ \frac{96 m^8+18 x \lambda m^4-x^2 \lambda^2}{96 m^{10}+66 x \lambda m^6} & \frac{72 m^8+12 x \lambda m^4-x^2 \lambda^2}{72 m^{10}+48 x \lambda m^6-x^2 \lambda^2 m^2} & \frac{m^2 \left(360 m^8+468 x \lambda m^4+97 x^2 \lambda^2\right)}{360 m^{12}+648 x \lambda m^8+301 x^2 \lambda^2 m^4+17 x^3 \lambda^3} \\ \frac{4752 m^{12}+888 x \lambda m^8-48 x^2 \lambda^2 m^4-x^3 \lambda^3}{4752 m^{14}+3264 x \lambda m^{10}} & \frac{360 m^{12}+1692 x \lambda m^8+301 x^2 \lambda^2 m^4-17 x^3 \lambda^3}{m^6 \left(360 m^8+1872 x \lambda m^4+1117 x^2 \lambda^2\right)} & \frac{6120 m^{12}+5220 x \lambda m^8+1193 x^2 \lambda^2 m^4+38 x^3 \lambda^3}{6120 m^{14}+8280 x \lambda m^{10}+3293 x^2 \lambda^2 m^6+327 x^3 \lambda^3 m^2} \end{pmatrix}$$

```
BPG[a_, b_] :=
 BPG[a, b] = LaplaceTransform[BorelPade[a, b], x, s] /. s → 1 // Simplify //
   FullSimplify[#, Assumptions → {λ > 0, m > 0}] &
```

```
Table[BPG[a, 1], {a, 0, 3}]
```

$$\left\{\frac{2\,m^2\,e^{\frac{2 m^4}{\lambda}}\,\Gamma\!\left(0,\frac{2 m^4}{\lambda}\right)}{\lambda},\ \frac{2-\frac{9\,m^4\,e^{\frac{3 m^4}{2\lambda}}\,\mathrm{Ei}\left(-\frac{3 m^4}{2\lambda}\right)}{\lambda}}{8\,m^2},\ \frac{-\frac{8192\,m^8\,e^{\frac{16 m^4}{11\lambda}}\,\mathrm{Ei}\left(-\frac{16 m^4}{11\lambda}\right)}{\lambda}-121\,\lambda+2354\,m^4}{7986\,m^6},\right.$$
$$\left.-\left(\left(1\,056\,655\,611\,m^{12}\,e^{\frac{99 m^4}{68\lambda}}\,\mathrm{Ei}\!\left(-\frac{99\,m^4}{68\,\lambda}\right)+68\,\lambda\left(9248\,\lambda^2-4\,419\,447\,m^8+215\,220\,\lambda\,m^4\right)\right)\Big/\left(1\,026\,306\,048\,\lambda\,m^{10}\right)\right)\right\}$$

```
ClearAll[NBPG, NPGs]

NBPG[a_, b_] := NBPG[a, b] =
  Table[{λ, LaplaceTransform[BorelPade[a, b], x, s] /. s → 1 /. m → 1}, {λ, 0, 5, 0.1}]

(*NBPG[1,1]=
  Table[{λ,NIntegrate[Exp[-x]BorelPade[1,2]/.m→1,{x,0,∞},WorkingPrecision→30]},
   {λ,0,5,0.1}];*)
NBPG[2, 1] = Table[{λ, NIntegrate[Exp[-x] BorelPade[2, 2] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 45]}, {λ, 0, 5, 0.1}];
NBPG[3, 1] = Table[{λ, NIntegrate[Exp[-x] BorelPade[3, 2] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 30]}, {λ, 0, 5, 0.1}];
NBPG[1, 2] = Table[{λ, NIntegrate[Exp[-x] BorelPade[1, 2] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 30]}, {λ, 0, 5, 0.1}];
NBPG[2, 2] = Table[{λ, NIntegrate[Exp[-x] BorelPade[2, 2] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 45]}, {λ, 0, 5, 0.1}];
NBPG[3, 2] = Table[{λ, NIntegrate[Exp[-x] BorelPade[3, 2] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 30]}, {λ, 0, 5, 0.1}];
NBPG[1, 3] = Table[{λ, NIntegrate[Exp[-x] BorelPade[1, 3] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 30]}, {λ, 0, 5, 0.1}];
NBPG[2, 3] = Table[{λ, NIntegrate[Exp[-x] BorelPade[2, 3] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 30]}, {λ, 0, 5, 0.1}];
NBPG[3, 3] = Table[{λ, NIntegrate[Exp[-x] BorelPade[3, 3] /. m → 1,
      {x, 0, ∞}, WorkingPrecision → 30]}, {λ, 0, 5, 0.1}];
```

NIntegrate::precw : The precision of the argument function ($1.\,e^{-x}$) is less than WorkingPrecision (45.). ≫

NIntegrate::precw :

The precision of the argument function ($\frac{e^{-x}\left(1+0.0166667\,x-0.000138889\,x^2\right)}{1+0.0666667\,x-0.000138889\,x^2}$) is less than WorkingPrecision (45.). ≫

NIntegrate::precw :

The precision of the argument function ($\frac{e^{-x}\left(1+0.0333333\,x-0.000555556\,x^2\right)}{1+0.133333\,x-0.000555556\,x^2}$) is less than WorkingPrecision (45.). ≫

General::stop : Further output of NIntegrate::precw will be suppressed during this calculation. ≫

NIntegrate::ncvb : NIntegrate failed to converge to prescribed accuracy after
9 recursive bisections in x near $\{x\}$ = {75.1489285231439676997156244380075988512619334}.
NIntegrate obtained 0.817350542166047517630099509788974127068805045 and
$5.290421043000050537114218682772897709705633316\times10^{-33}$ for the integral and error estimates. ≫

NIntegrate::ncvb : NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x near $\{x\}$ = {68.0079526086468872620272154053242968119309705}. NIntegrate obtained 0.798202524039832931984313148207195779211279388 and $9.44027351128877318557846137330198559451747553\times10^{-29}$ for the integral and error estimates. >>

NIntegrate::ncvb : NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x near $\{x\}$ = {56.8807320667301246894755751938663576579826771}. NIntegrate obtained 0.780690780772040543747074919531026963953427732 and $1.61918011095201769088758421592912547028083897\times10^{-24}$ for the integral and error estimates. >>

General::stop : Further output of NIntegrate::ncvb will be suppressed during this calculation. >>

NIntegrate::slwcon : Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. >>

NIntegrate::slwcon : Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. >>

NIntegrate::slwcon : Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. >>

General::stop : Further output of NIntegrate::slwcon will be suppressed during this calculation. >>

NIntegrate::precw : The precision of the argument function ($1.\,e^{-x}$) is less than WorkingPrecision (30.). >>

NIntegrate::precw : The precision of the argument function ($\frac{e^{-x}\left(1+0.47\,x+0.00836111\,x^2-0.0000472222\,x^3\right)}{1+0.52\,x+0.0310278\,x^2}$) is less than WorkingPrecision (30.). >>

NIntegrate::precw : The precision of the argument function ($\frac{e^{-x}\left(1+0.94\,x+0.0334444\,x^2-0.000377778\,x^3\right)}{1+1.04\,x+0.124111\,x^2}$) is less than WorkingPrecision (30.). >>

General::stop : Further output of NIntegrate::precw will be suppressed during this calculation. >>

NIntegrate::precw : The precision of the argument function ($1.\,e^{-x}$) is less than WorkingPrecision (30.). >>

NIntegrate::precw : The precision of the argument function ($\frac{e^{-x}\,(1+0.025\,x)}{1+0.075\,x+0.000416667\,x^2}$) is less than WorkingPrecision (30.). >>

NIntegrate::precw : The precision of the argument function ($\frac{e^{-x}\,(1+0.05\,x)}{1+0.15\,x+0.00166667\,x^2}$) is less than WorkingPrecision (30.). >>

General::stop : Further output of NIntegrate::precw will be suppressed during this calculation. >>

NIntegrate::precw : The precision of the argument function ($1.\,e^{-x}$) is less than WorkingPrecision (45.). >>

NIntegrate::precw : The precision of the argument function ($\frac{e^{-x}\left(1+0.0166667\,x-0.000138889\,x^2\right)}{1+0.0666667\,x-0.000138889\,x^2}$) is less than WorkingPrecision (45.). >>

NIntegrate::precw : The precision of the argument function ($\frac{e^{-x}\left(1+0.0333333\,x-0.000555556\,x^2\right)}{1+0.133333\,x-0.000555556\,x^2}$) is less than WorkingPrecision (45.). >>

General::stop : Further output of NIntegrate::precw will be suppressed during this calculation. >>

NIntegrate::ncvb : NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x near $\{x\}$ = {75.1489285231439676997156244380075988512619334}. NIntegrate obtained 0.817350542166047517630099509788974127068805045 and $5.29042104300005053711421868277289770970563316\times10^{-33}$ for the integral and error estimates. >>

NIntegrate::ncvb : NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x near $\{x\}$ = {68.0079526086468872620272154053242968119309705}. NIntegrate obtained 0.798202524039832931984313148207195779211279388 and $9.44027351128877318557846137330198559451747553\times10^{-29}$ for the integral and error estimates. >>

NIntegrate::ncvb : NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in x near $\{x\}$ = {56.8807320667301246894755751938663576579826771}. NIntegrate obtained 0.780690780772040543747074919531026963953427732 and $1.61918011095201769088758421592912547028083897\times10^{-24}$ for the integral and error estimates. >>

General::stop : Further output of NIntegrate::ncvb will be suppressed during this calculation. >>

NIntegrate::slwcon : Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. >>

NIntegrate::slwcon : Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. >>

NIntegrate::slwcon : Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. >>

General::stop : Further output of NIntegrate::slwcon will be suppressed during this calculation. >>

NIntegrate::precw : The precision of the argument function ($1.\,e^{-x}$) is less than WorkingPrecision (30.). >>

NIntegrate::precw : The precision of the argument function ($\frac{e^{-x}\left(1+0.47\,x+0.00836111\,x^2-0.0000472222\,x^3\right)}{1+0.52\,x+0.0310278\,x^2}$) is less than WorkingPrecision (30.). >>

NIntegrate::precw : The precision of the argument function ($\frac{e^{-x}\left(1+0.94\,x+0.0334444\,x^2-0.000377778\,x^3\right)}{1+1.04\,x+0.124111\,x^2}$) is less than WorkingPrecision (30.). >>

General::stop : Further output of NIntegrate::precw will be suppressed during this calculation. >>

NIntegrate::precw : The precision of the argument function ($1.\,e^{-x}$) is less than WorkingPrecision (30.). >>

NIntegrate::precw : The precision of the argument function ($\frac{e^{-x}\,(1+0.0222222\,x)}{1+0.0722222\,x+0.000277778\,x^2+2.31481\times10^{-6}\,x^3}$) is less than WorkingPrecision (30.). >>

NIntegrate::precw : The precision of the argument function ($\frac{e^{-x}\,(1+0.0444444\,x)}{1+0.144444\,x+0.00111111\,x^2+0.0000185185\,x^3}$) is less than WorkingPrecision (30.). >>

General::stop : Further output of NIntegrate::precw will be suppressed during this calculation. >>

NIntegrate::precw : The precision of the argument function ($1.\,e^{-x}$) is less than WorkingPrecision (30.). >>

NIntegrate::precw :

The precision of the argument function ($\frac{e^{-x}\left(1 + 0.13\,x + 0.00269444\,x^2\right)}{1 + 0.18\,x + 0.00836111\,x^2 + 0.0000472222\,x^3}$) is less than WorkingPrecision (30.). ≫

NIntegrate::precw :

The precision of the argument function ($\frac{e^{-x}\left(1 + 0.26\,x + 0.0107778\,x^2\right)}{1 + 0.36\,x + 0.0334444\,x^2 + 0.000377778\,x^3}$) is less than WorkingPrecision (30.). ≫

General::stop : Further output of NIntegrate::precw will be suppressed during this calculation. ≫

NIntegrate::precw : The precision of the argument function ($1.\,e^{-x}$) is less than WorkingPrecision (30.). ≫

NIntegrate::precw : The precision of the argument function

($\frac{e^{-x}\left(1 + 0.0852941\,x + 0.00194935\,x^2 + 6.20915\times10^{-6}\,x^3\right)}{1 + 0.135294\,x + 0.00538072\,x^2 + 0.0000534314\,x^3}$) is less than WorkingPrecision (30.). ≫

NIntegrate::precw : The precision of the argument function

($\frac{e^{-x}\left(1 + 0.170588\,x + 0.00779739\,x^2 + 0.0000496732\,x^3\right)}{1 + 0.270588\,x + 0.0215229\,x^2 + 0.000427451\,x^3}$) is less than WorkingPrecision (30.). ≫

General::stop : Further output of NIntegrate::precw will be suppressed during this calculation. ≫

```
(*Table[NBPG[a,b],{a,3},{b,3}]*)

NPGs = Table[{a, b, NBPG[a, b]}, {a, 3}, {b, 3}];

plotdata = ({{#[[ ;; , 1]], #[[ ;; , 2]] / Thread[Gexact /. m → 1 /. λ → #[[ ;; , 1]]]} // Transpose} & /@
    Flatten[NPGs, 1][[ ;; , 3]])[[ ;; , 1, ;; , ;;]];
```

Power::infy : Infinite expression $\frac{1}{0.}$ encountered. ≫

Power::infy : Infinite expression $\frac{1}{0.}$ encountered. ≫

Power::infy : Infinite expression $\frac{1}{0.}$ encountered. ≫

General::stop : Further output of Power::infy will be suppressed during this calculation. ≫

```
Show[Plot[1, {λ, 0, 5}, PlotRange → {0.9, 1.06}, AxesLabel → {λ, "Gapprox/G̅"},
  PlotStyle → {Black, Thick}, PlotLegends → {"G̅"}], ListPlot[plotdata, PlotLegends →
   ({Subsuperscript["Borel-Pade", ToString[#[[2]]], ToString[#[[1]]]] & /@
       Flatten[NPGs, 1][[ ;; , 1 ;; 2]]} // Flatten),
  PlotMarkers → {Automatic, Medium}], LabelStyle → Medium]
```




```
(*Export["Gbar-PadeBorel.pdf",%132]*)
```

Gbar–PadeBorel.pdf

```
BPσs = Table[{a, b, InverseLaplaceTransform[BorelPade[a, b], x, v] /. v → 1},
   {a, 3}, {b, 3}] // FullSimplify[#, Assumptions → λ > 0] &
```

$$\begin{pmatrix} \left\{1,\ 1,\ \frac{9\, e^{-\frac{3 m^4}{2\lambda}}\, m^2}{8\lambda}\right\} & \left\{1,\ 2,\ \frac{\sqrt{\frac{3}{19}}\, e^{-\frac{\left(9+\sqrt{57}\right) m^4}{\lambda}} \left(5+\sqrt{57}+\left(-5+\sqrt{57}\right) e^{\frac{2\sqrt{57}\, m^4}{\lambda}}\right) m^2}{\lambda}\right\} \\ \left\{2,\ 1,\ \frac{4096\, e^{-\frac{16 m^4}{11\lambda}}\, m^2}{3993\lambda}\right\} & \left\{2,\ 2,\ \frac{12\, e^{\frac{24 m^4}{\lambda}}\, m^2 \left(3\cosh\left(\frac{18\sqrt{2}\, m^4}{\lambda}\right)+2\sqrt{2}\,\sinh\left(\frac{18\sqrt{2}\, m^4}{\lambda}\right)\right)}{\lambda}\right\} \\ \left\{3,\ 1,\ \frac{352\,218\,537\, e^{-\frac{99 m^4}{68\lambda}}\, m^2}{342\,102\,016\lambda}\right\} & \left\{3,\ 2,\ \frac{9\sqrt{\frac{2}{6583}}\, e^{-\frac{6\left(156+\sqrt{13\,166}\right) m^4}{1117\lambda}} \left(4\,554\,518\,636+39\,693\,141\sqrt{13\,166}+\left(-4\,554\,518\,636+39\,693\,141\sqrt{13\,166}\right) e^{\frac{12\sqrt{13\,166}\, m^4}{1117\lambda}}\right) m^2}{1\,393\,668\,613\lambda}\right\} \end{pmatrix}$$

```
Subsuperscript["Borel-Pade", ToString[#[[2]]], ToString[#[[1]]]] & /@
 Flatten[BPσs, 1][[ ;; , 1 ;; 2]]
```

{Borel–Pade$^1_1$, Borel–Pade$^1_2$, Borel–Pade$^1_3$, Borel–Pade$^2_1$,
Borel–Pade$^2_2$, Borel–Pade$^2_3$, Borel–Pade$^3_1$, Borel–Pade$^3_2$, Borel–Pade$^3_3$}

```
Plot[Re@{1, Flatten[BPσs, 1][[ ;; , 3]] / σexact /. m → 1} // Flatten // Evaluate,
 {λ, 0, 20}, PlotLegends →
  ({σ, σ[Subsuperscript["Borel-Pade", ToString[#[[2]]], ToString[#[[1]]]]] & /@
      Flatten[BPσs, 1][[ ;; , 1 ;; 2]]} // Flatten), AxesLabel → {λ, "σapprox / σexact"},
 PlotRange → {-1, 3}, PlotStyle → ({{Automatic, Thick}, Dotted, Dashing[Large],
    DotDashed, Dashing[{Small, Medium}], Dashing[{0, Small, Tiny}],
    Dashing[{Medium, Medium, 0}], DotDashed, {Dotted, Thick}}), LabelStyle → Medium]
```

N::meprec : Internal precision limit \$MaxExtraPrecision = 50. reached while evaluating $39693141 \sqrt{13166}$. ≫




```
(*Export["sigma-PadeBorel.pdf",%134]*)
```

sigma−PadeBorel.pdf

```
Plot[{σexact,
    (HeavisideTheta[λ] InverseLaplaceTransform[BorelPade[1, 1], x, v] /. v → 1)} /.
  m → 1 // Evaluate, {λ, 0, 10}, PlotStyle → {DotDashed, Dashed}]
```




```
FindRoot[Re[BPG[0, 1]] /. m → 1, {λ, -5}]
```

{λ → −5.36902}

```
BPG[0, 1] /. m → 1 /. {λ → -5.36902}
```

$4.02578\times 10^{-17} + 0.806319\, i$

2 PI Approximations

```
Clear[Γ2, G]
```

$$\texttt{Γ2[G\_] := Γ2[G] =} \ \frac{1}{2}\,\mathrm{Log}\left[G^{-1}\right] + \frac{1}{2}\, m^2\, G + \gamma 2[G]$$

To determine γ2[G] we expand the left and right hand sides of the G equation of motion to $O(G^n)$:

$$\texttt{GeomExpansion[n\_] := Module}\left[\left\{\gamma 2 = \mathrm{Sum}\left[\gamma[i]\ \left(\lambda G^2\right)^i,\ \{i, 1, n\}\right]\right\},\ \mathrm{Series}\left[\left\{\mathrm{Gexact},\ \frac{1}{m^2 + 2\,\mathrm{D}[\gamma 2, G]}\ /.\ G \to \mathrm{Gexact}\right\},\ \{\lambda, 0, n\}\right]\ //\ \mathrm{Normal}\right]$$

and match coefficients:

```
nthMatchingEqs[n_] := Thread[CoefficientList[GeomExpansion[n][[1]], λ] ==
   CoefficientList[GeomExpansion[n][[2]], λ]]
```

```
GeomExpansion[3]
```

$$\left\{-\frac{11\,\lambda^3}{8\,m^{14}} + \frac{2\,\lambda^2}{3\,m^{10}} - \frac{\lambda}{2\,m^6} + \frac{1}{m^2},\right.$$
$$\left. -\frac{4\left(48\,\gamma(1)^3 + 12\,\gamma(1)^2 - 48\,\gamma(2)\,\gamma(1) + 2\,\gamma(1) - 9\,\gamma(2) + 9\,\gamma(3)\right)\lambda^3}{3\,m^{14}} + \frac{2\left(8\,\gamma(1)^2 + \gamma(1) - 4\,\gamma(2)\right)\lambda^2}{m^{10}} - \frac{4\,\gamma(1)\,\lambda}{m^6} + \frac{1}{m^2}\right\}$$

```
γ2coeffs = Solve[nthMatchingEqs[20], Table[γ[i], {i, 20}]][[1]];
```

$$\mathrm{Sum}\left[\gamma[i]\ \left(\lambda G^2\right)^i,\ \{i, 1, 20\}\right]\ /.\ \gamma 2\mathrm{coeffs}$$

$$-\frac{1\,134\,464\,795\,789\,402\,074\,081\,G^{40}\,\lambda^{20}}{679\,477\,248} + \frac{70\,482\,935\,423\,952\,918\,475\,G^{38}\,\lambda^{19}}{537\,919\,488} - \frac{921\,138\,067\,678\,697\,395\,G^{36}\,\lambda^{18}}{84\,934\,656} + \frac{4\,231\,429\,245\,358\,235\,G^{34}\,\lambda^{17}}{4\,456\,448} - \frac{1\,665\,014\,342\,007\,385\,G^{32}\,\lambda^{16}}{18\,874\,368} + \frac{5\,148\,422\,676\,667\,G^{30}\,\lambda^{15}}{589\,824} - \frac{11\,440\,081\,763\,125\,G^{28}\,\lambda^{14}}{12\,386\,304} + \frac{201\,268\,707\,239\,G^{26}\,\lambda^{13}}{1\,916\,928} - \frac{3\,802\,117\,511\,G^{24}\,\lambda^{12}}{294\,912} + \frac{697\,775\,057\,G^{22}\,\lambda^{11}}{405\,504} - \frac{5\,151\,939\,G^{20}\,\lambda^{10}}{20\,480} + \frac{374\,755\,G^{18}\,\lambda^{9}}{9216} - \frac{271\,217\,G^{16}\,\lambda^{8}}{36\,864} + \frac{8143\,G^{14}\,\lambda^{7}}{5376} - \frac{93\,G^{12}\,\lambda^{6}}{256} + \frac{101\,G^{10}\,\lambda^{5}}{960} - \frac{5\,G^{8}\,\lambda^{4}}{128} + \frac{G^{6}\,\lambda^{3}}{48} - \frac{G^{4}\,\lambda^{2}}{48} + \frac{G^{2}\,\lambda}{8}$$

The series has super-exponential coefficients, aymptotically the same as perturbation theory up to a constant factor:

```
Table[{i, γ[i]/(.03 (-2/3)^(i-1) (i - 1) !)}, {i, 20}] /. γ2coeffs // TableForm // N
```

1.	4.16667
2.	1.04167
3.	0.78125
4.	0.732422
5.	0.739746
6.	0.766296
7.	0.798765
8.	0.831384
9.	0.861575
10.	0.888337
11.	0.911483
12.	0.931236
13.	0.947998
14.	0.962215
15.	0.974316
16.	0.984677
17.	0.993614
18.	1.00139
19.	1.0082
20.	1.01422

```
Clear[Gsolns]
```

```
Gsolns[n_] := Gsolns[n] =
  G /. Solve[0 == D[1/2 Log[G^-1] + 1/2 m^2 G + Sum[γ[i] (λ G^2)^i, {i, 1, n}], G] /. γ2coeffs, G] //
   FullSimplify[#, Assumptions → {m > 0, λ > 0}] &
```

```
Gsolns[1]
```

$$\left\{-\frac{\sqrt{2\lambda + m^4} + m^2}{\lambda}, \frac{\sqrt{2\lambda + m^4} - m^2}{\lambda}\right\}$$

```
Gsolns[3]
```

$\{\text{Root}[3\,\#1^6\,\lambda^3 - 2\,\#1^4\,\lambda^2 + 6\,\#1^2\,\lambda + 12\,\#1\,m^2 - 12\,\&, 1], \text{Root}[3\,\#1^6\,\lambda^3 - 2\,\#1^4\,\lambda^2 + 6\,\#1^2\,\lambda + 12\,\#1\,m^2 - 12\,\&, 2],$
$\text{Root}[3\,\#1^6\,\lambda^3 - 2\,\#1^4\,\lambda^2 + 6\,\#1^2\,\lambda + 12\,\#1\,m^2 - 12\,\&, 3], \text{Root}[3\,\#1^6\,\lambda^3 - 2\,\#1^4\,\lambda^2 + 6\,\#1^2\,\lambda + 12\,\#1\,m^2 - 12\,\&, 4],$
$\text{Root}[3\,\#1^6\,\lambda^3 - 2\,\#1^4\,\lambda^2 + 6\,\#1^2\,\lambda + 12\,\#1\,m^2 - 12\,\&, 5], \text{Root}[3\,\#1^6\,\lambda^3 - 2\,\#1^4\,\lambda^2 + 6\,\#1^2\,\lambda + 12\,\#1\,m^2 - 12\,\&, 6]\}$

```
Series[Gsolns[1], {λ, 0, 1}] // FullSimplify[#, Assumptions → {m > 0, λ > 0}] &
```

$$\left\{-\frac{2m^2}{\lambda} - \frac{1}{m^2} + \frac{\lambda}{2m^6} + O(\lambda^2), \frac{1}{m^2} - \frac{\lambda}{2m^6} + O(\lambda^2)\right\}$$

```
g2loop = Gsolns[1][[2]]
```

$$\frac{\sqrt{2\lambda + m^4} - m^2}{\lambda}$$

```
Plot[{Gexact, g2loop, Gsolns[2][[2]]} /. m → 1 // Evaluate, {λ, -5, 5}, PlotLegends →
  {G̅, Subsuperscript[G, 1, "(2)"], Subsuperscript[G, 2, "(2)"]}, PlotRange → {0, 2}]
```



```
Series[(Gsolns[2][[1]] /. λ → λ + ⅈ ϵ) - (Gsolns[2][[1]] /. λ → λ - ⅈ ϵ) /. m → 1,
 {ϵ, 0, 0}, Assumptions → {λ < -5}]
```

$$\frac{1}{2\lambda}\left(\sqrt{\left(-\sqrt[3]{-2\,i\sqrt{50\lambda^2-207\lambda-81}+23\lambda+18}\,\left(\lambda^{2/3}\right)^{*}-\frac{9\left(\lambda^{4/3}\right)^{*}}{\sqrt[3]{-2\,i\sqrt{50\lambda^2-207\lambda-81}+23\lambda+18}}-12\Big/\left(\sqrt{\left(\sqrt[3]{-2\,i\sqrt{50\lambda^2-207\lambda-81}+23\lambda+18}\,\left(\frac{1}{\lambda^{4/3}}\right)^{*}+\frac{9\left(\frac{1}{\lambda^{2/3}}\right)^{*}}{\sqrt[3]{-2\,i\sqrt{50\lambda^2-207\lambda-81}+23\lambda+18}}+\frac{2}{\lambda}\right)}\right)+4\lambda\right)}+\lambda\sqrt{\left(\sqrt[3]{-2\,i\sqrt{50\lambda^2-207\lambda-81}+23\lambda+18}\,\left(\frac{1}{\lambda^{4/3}}\right)^{*}+\frac{9}{\sqrt[3]{23\lambda^3-2\,i\sqrt{50\lambda^2-207\lambda-81}\,\lambda^2+18\lambda^2}}+\frac{2}{\lambda}\right)}-\sqrt{\left(\frac{9}{\sqrt[3]{\lambda^2\left(2\sqrt{-50\lambda^2+207\lambda+81}+23\lambda+18\right)}}+\frac{\sqrt[3]{2\sqrt{-50\lambda^2+207\lambda+81}+23\lambda+18}}{\lambda^{4/3}}+\frac{2}{\lambda}\right)}\,\lambda-\sqrt{\left(-\frac{9\lambda^{4/3}}{\sqrt[3]{2\sqrt{-50\lambda^2+207\lambda+81}+23\lambda+18}}-\sqrt[3]{2\sqrt{-50\lambda^2+207\lambda+81}+23\lambda+18}\,\lambda^{2/3}-12\Big/\left(\sqrt{\left(\frac{\sqrt[3]{2\sqrt{-50\lambda^2+207\lambda+81}+23\lambda+18}}{\lambda^{4/3}}+\frac{9}{\lambda^{2/3}\sqrt[3]{2\sqrt{-50\lambda^2+207\lambda+81}+23\lambda+18}}+\frac{2}{\lambda}\right)}\right)+4\lambda\right)}\right)+O\left(\epsilon^{1}\right)$$

```
(-1/(2 π ⅈ) Series[(g2loop /. λ → λ + ⅈ ϵ) - (g2loop /. λ → λ - ⅈ ϵ) /. m → 1,
      {ϵ, 0, 0}, Assumptions → {λ < -1/2}] //
    FullSimplify[#, Assumptions → {λ < -1/2}] & // Normal) /. λ → -λ
```

$$\frac{\sqrt{2\lambda - 1}}{\pi\lambda}$$

```
σ2 = Sqrt[2 λ - m^4]/(π λ) HeavisideTheta[λ - m^4/2]
```

$$\frac{\sqrt{2\lambda - m^4}\,\theta\left(\lambda - \frac{m^4}{2}\right)}{\pi\lambda}$$

```
Plot[{σexact, σ2} /. m → 1 // Evaluate, {λ, 0, 10}, FrameLabel → {λ, σ},
 PlotTheme → "Scientific", PlotLegends → {"σ", "σ(1)"}, PlotRange → {0, 1/3},
 PlotStyle → {Automatic, Dashed}, PlotPoints → 50, LabelStyle → Medium]
```




```
(*Export["sigma-2pi-2loop.pdf",%137]*)
```

```
Plot[Re /@ {Gexact, Gpert[1], PadeApproximant[Gpert[1], {λ, 0, {0, 1}}],
     BPG[0, 1], g2loop} /. m → 1 // Evaluate, {λ, -6, 6}, PlotRange → {-.5, 2.5},
 PlotLegends → {"Exact", "1st Order Pert.", Subsuperscript["Pade", 1, 0],
    Subsuperscript["Borel-Pade", 1, 0], "1st Order 2PI"}, AxesLabel → {λ, "Re[G̅]"},
 PlotStyle → {Thick, Dashed, Dashing[{Small, Medium, Large}], Thin, DotDashed},
 LabelStyle → Medium]
```




```
(*Export["Gbar-all-first-order-approxs.pdf",%139]*)
```